

### **Description of the OMM file (version 0) and several DELPHI6 procedures to use a map defined by this OMM file**

In a general way, to use a map, two files are necessary : the picture itself and another file describing the map.

Let's suppose that the mantissa of both files be "MANTISSA\_OF\_20\_CHARA". This mantissa must be 20 characters long maximum. In this case, the OMMMap program will supply the two files :

- MANTISSA\_OF\_20\_CHARA.PNG (the picture of the map on 800x600 pixels, graphic format which is compulsory)
- MANTISSA\_OF\_20\_CHARA.OMM (the description of this map).

The MANTISSA\_OF\_20\_CHARA.OMM file, contains, on only one line:

- The format version of this OMM file on one character followed by a space, so "0 " for this first version. The possible following versions will be backward compatible with this first version, by adding pieces of information after the 30 characters maximum of the comment (and not before), so from the position 75 (included).
- The zoom of the Openstreetmap map (displayed by OMMMap), on 2 figures ("02" to "19") followed by a space.
- The contraction factor along X on 8 characters and 5 decimals, followed by a space, as for example: "01.23456 "). The standard would be 01.00000 (i.e. no contraction).
- The contraction factor along Y on 8 characters and 5 decimals, followed by a space, as for example: "05.01010 ". The standard would be 01.00000 (i.e. no contraction).
- The latitude of the center of the map in decimal degree at the format "+/-" + degrees on 2 figures + "." + 5 decimals. Examples: "+49.54280" (north of the equator) or "-25.12300" (south of the equator). The limit on the latitude is equal to +/-85.05000°. The latitude is followed by a space.
- The longitude of the center of the map in decimal degree at the format "+/-" + degrees on 3 figures + "." + 5 decimals. Examples: "+003.54280" (east of the Greenwich meridian) or "-025.12345" (west of the Greenwich meridian). The limit on the longitude is equal to +/-180.00000°. The longitude is followed by a space.
- The possible comment of 30 characters maximum.

In the ZIP file, two files are proposed as examples: EUROPE.OMM and EUROPE.PNG.

Several DELPHI6 procedures to use this OMM file can be of some use, for the programmer.

Note: below, according to the IMAGE\_APRS\_800\_600 variable, Multipsk can display the map on the original dimensions (800x600 pixels) but also in a reduced format (480x360 pixels).

```
{Public variables relatives to the OMM map. V.4.48.4}
VAR XC_frac,YC_frac:DOUBLE;//fractional positions at the center of the map
VAR LE_MAP,HE_MAP:WORD;//length and height of the map in pixels (a priori: 800 x 600)
VAR LC_COMPLETE_MAP,HC_COMPLETE_MAP:DWORD;//length and height of the complete map for a given zoom (example 1024x1024 at zoom 2)
VAR CONTRACTION_FACTOR_ALONG_X,CONTRACTION_FACTOR_ALONG_Y:DOUBLE;

{*****}

PROCEDURE CALCULATION_OF_THE_FRACTIONAL_POSITIONS_AT_THE_CENTER_OF_THE_MAP(LONGITUDE,LATITUDE:DOUBLE);//V.4.48.4
{Calculation of the public variables XC_frac and YC_frac, from the latitude and longitude coordinates (in decimal degrees) at the center of the map}
VAR X_Pi,Y_Pi:DOUBLE;//in radians
BEGIN
  X_Pi:=LONGITUDE*Pi/180;
  Y_Pi:=arcsinh(tan(LATITUDE*Pi/180));
  XC_frac:=(1+X_Pi/Pi)/2;
  YC_frac:=(1-Y_Pi/Pi)/2;
END;

{*****}
FUNCTION CALCULATION_OF_THE_DECIMAL_LONGITUDE_AT_AN_X_PIXEL_POSITION_ON_THE_MAP(XM:DWORD):DOUBLE;//V.4.48.4
{From the pixel position XM between 0..LE_MAP-1 (so 0..799 for standard map or 0..479 for small map(480x360)), position called "M"}
VAR XM_frac,XM_Pi:DOUBLE;
VAR TERM:DOUBLE;
BEGIN
  IF LE_MAP<800 {réduit} THEN XM:=(XM * 800) DIV 480;//for the small map (480x360)
  TERM:=(XM-(800/2)+0.5)*CONTRACTION_FACTOR_ALONG_X;
  XM_frac:=XC_frac+TERM/LC_COMPLETE_MAP;
  {WRITELN('XM ',XM,' ',XC_frac:6:5,' ',XM_frac:6:5,' ',CONTRACTION_FACTOR_ALONG_X:5:5,' ',TERM:6:5);}
  XM_Pi:=Pi*(2*XM_frac-1);
  CALCULATION_OF_THE_DECIMAL_LONGITUDE_AT_AN_X_PIXEL_POSITION_ON_THE_MAP:=XM_Pi*180/Pi;{-180 to 180°}
END;

{*****}
```

```

FUNCTION CALCULATION_OF_THE_DECIMAL_LATITUDE_AT_AN_Y_PIXEL_POSITION_ON_THE_MAP(YM:DWORD):DOUBLE;//V.4.48.4
{From the pixel position YM between 0..HE_MAP-1 (so 0..599 for standard map or 0..359 for small map(480x360)), position called "M"}
VAR YM_frac, YM_Pi:DOUBLE;
VAR TERM:DOUBLE;
BEGIN
  IF HE_MAP<600 {réduit} THEN YM:=(YM * 600) DIV 360;//for the small map (480x360)
  TERM:=(YM-(600/2)+0.5)*CONTRACTION_FACTOR_ALONG_Y;
  YM_frac:=YC_frac+TERM/HC_COMPLETE_MAP;
  {WRITELN('YM ',YM,' ',YC_frac:6:5,' ',YM_frac:6:5,' ',CONTRACTION_FACTOR_ALONG_Y:5:5,' ',TERM:6:5);}
  YM_Pi:=Pi*(1-2*YM_frac);
  CALCULATION_OF_THE_DECIMAL_LATITUDE_AT_AN_Y_PIXEL_POSITION_ON_THE_MAP:=Arctan(sinh(YM_Pi))*180/Pi;{-85.051 to -85.051°}
END;

{*****}

PROCEDURE DETERMINATION_OF_THE_MOUSE_POSITION_IN_PIXELS_OVER_THE_OMM_MAP_FROM_THE_COORDINATES(LONGITUDE,LATITUDE:DOUBLE;VAR XM,YM:LONGINT);
{Inverse of CALCULATION_OF_THE_DECIMAL_LONGITUDE_AT_AN_X_PIXEL_POSITION_ON_THE_MAP +
CALCULATION_OF_THE_DECIMAL_LATITUDE_AT_AN_Y_PIXEL_POSITION_ON_THE_MAP
From the coordinates (longitude/latitude, it is determined the pixel position:
XM between 0..LE_MAP-1 (so 0..799 by default) / YM between 0..HE_MAP-1 (so 0..599 by default). The position called "M"}
VAR DXM_frac,DYM_frac,XM_frac,XM_Pi,YM_frac,YM_Pi,TERM:DOUBLE;
BEGIN
  XM_Pi:=LONGITUDE*Pi/180;
  XM_frac:=(1+XM_Pi/Pi)/2;
  DXM_frac:=XM_frac-XC_frac;
  TERM:=(800/2-0.5);
  XM:=ROUND(DXM_frac*LC_COMPLETE_MAP/CONTRACTION_FACTOR_ALONG_X+TERM);
  IF XM<0 THEN XM:=0; IF XM>800-1 THEN XM:=800-1;

  YM_Pi:=arcsinh(tan(LATITUDE*Pi/180));
  YM_frac:=(1-YM_Pi/Pi)/2;
  DYM_frac:=YM_frac-YC_frac;
  TERM:=(600/2-0.5);
  YM:=ROUND(DYM_frac*HC_COMPLETE_MAP/CONTRACTION_FACTOR_ALONG_Y+TERM);
  IF YM<0 THEN YM:=0; IF YM>600-1 THEN YM:=600-1;
  {WRITELN('Res ',LE_MAP,' / ',XM,' ',YM,' ',LONGITUDE:11,' ',LATITUDE:11);}
END;

{*****}

PROCEDURE
INITIALIZATION_OF_THE_OMM_MAP_PARAMETERS(LATITUDE_AT_THE_CENTER_OF_THE_MAP,LONGITUDE_AT_THE_CENTER_OF_THE_MAP:DOUBLE;CURRENT_ZOOM:BYTE);
{calculation of the public variables XC_frac and YC_frac, LE_MAP, HE_MAP, LC_COMPLETE_MAP and HC_COMPLETE_MAP}
{!!!!!!!!!!!!}
FUNCTION TWO_POWER_COUNTER(COUNTER:BYTE):DWORD;
{COUNTER between 0 and 31 for a DWORD}
VAR RESULTAT:DWORD;
VAR COMPTE:LONGINT;
BEGIN
  IF COUNTER=0 THEN RESULTAT:=1 ELSE

```

```

BEGIN
  RESULTAT:=1;
  FOR COMPTE:=1 TO COUNTER DO RESULTAT:=RESULTAT*2;
END;
TWO_POWER_COUNTER:=RESULTAT;
END;
BEGIN
  {Calculation of the public variables XC_frac and YC_frac, from the latitude and longitude coordinates (in decimal degrees) at the center of the
map}
  CALCULATION_OF_THE_FRACTIONAL_POSITIONS_AT_THE_CENTER_OF_THE_MAP(LONGITUDE_AT_THE_CENTER_OF_THE_MAP,LATITUDE_AT_THE_CENTER_OF_THE_MAP);
  {Determination of the length and height of the map in pixels (a priori: 800 x 600 but can be 480x360 as proposed by Multipsk)}
  IF IMAGE_APRS_800_600=TRUE THEN
  BEGIN LE_MAP:=800; HE_MAP:=600;{normal} END ELSE BEGIN LE_MAP:=480; HE_MAP:=360;{reduced} END;
  {Determination of the length and height of the complete map for a given zoom (example 1024x1024 at zoom 2) from the CURRENT_ZOOM}
  LC_COMPLETE_MAP:=256*TWO_POWER_COUNTER(CURRENT_ZOOM); HC_COMPLETE_MAP:=LC_COMPLETE_MAP;
  {WRITELN('LE_MAP ',LE_MAP,' ',HE_MAP,' ',LC_COMPLETE_MAP);}
END;

```